(eval-when (:compile-toplevel :load-toplevel :execute)
  (sb-int:defconstant-eqx OPT
    #+swank '(optimize (speed 3) (safety 2))
    #-swank '(optimize (speed 3) (safety 0) (debug 0))
    #'equal)
  #+swank (ql:quickload '(:cl-debug-print :fiveam) :silent t)
  #-swank (set-dispatch-macro-character
           ;; enclose the form with VALUES to avoid being captured by LOOP macro
           #\# #\> (lambda (s c p) (declare (ignore c p)) `(values ,(read s nil nil t)))))
#+swank (cl-syntax:use-syntax cl-debug-print:debug-print-syntax)
#-swank (disable-debugger) ; for CS Academy

;; BEGIN_INSERTED_CONTENTS
(defmacro with-buffered-stdout (&body body)
  "Buffers all outputs to *STANDARD-OUTPUT* in BODY and flushes them to
*STANDARD-OUTPUT* after BODY has been done (without error). Note that only
BASE-CHAR is allowed."
  (let ((out (gensym)))
    `(let ((,out (make-string-output-stream :element-type 'base-char)))
       (let ((*standard-output* ,out))
         ,@body)
       (write-string (get-output-stream-string ,out)))))

(macrolet ((def (name fname)
             `(define-modify-macro ,name (new-value) ,fname)))
  (def minf min)
  (def maxf max)
  (def mulf *)
  (def divf /)
  (def iorf logior)
  (def xorf logxor)
  (def andf logand))

(defmacro dbg (&rest forms)
  #+swank
  (if (= (length forms) 1)
      `(format *error-output* "~A => ~A~%" ',(car forms) ,(car forms))
      `(format *error-output* "~A => ~A~%" ',forms `(,,@forms)))
  #-swank (declare (ignore forms)))

(defmacro define-int-types (&rest bits)
  `(progn
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "UINT~A" b)) () '(unsigned-byte ,b))) bits)
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "INT~A" b)) () '(signed-byte ,b))) bits)))
(define-int-types 2 4 7 8 15 16 31 32 62 63 64)

(declaim (inline println))
(defun println (obj &optional (stream *standard-output*))
  (let ((*read-default-float-format* 'double-float))
    (prog1 (princ obj stream) (terpri stream))))

(defconstant +mod+ 1000000007)

;;;
;;; Body
;;;

(defun main ()
  (let* ((n (read))
         (k (read))
         (c (read))
         (s (read-line))
         (dp1 (make-array (+ n 1) :element-type 'uint32 :initial-element 0))
         (dp2 (make-array (+ n 1) :element-type 'uint32 :initial-element 0))
         (counts (make-array (+ n 1) :element-type 'int32 :initial-element 0)))
    (when (and (zerop c) (> (count #\o s) k))
      (terpri)
      (return-from main))
    (loop for i from 1 to n
          do (if (char= (aref s (- i 1)) #\o)
                 (setf (aref dp1 i)
                       (max (aref dp1 (- i 1))
                            (if (>= (- i c 1) 0)
                                (+ 1 (aref dp1 (- i c 1)))
                                1)))
                 (setf (aref dp1 i) (aref dp1 (- i 1)))))
    (loop for i from (- n 1) downto 0
          do (if (char= (aref s i) #\o)
                 (setf (aref dp2 i)
                       (max (aref dp2 (+ i 1))
                            (if (<= (+ i c 1) n)
                                (+ 1 (aref dp2 (+ i c 1)))
                                1)))
                 (setf (aref dp2 i) (aref dp2 (+ i 1)))))
    (loop for end1 from (- c) to n
          for end2 from 0 to (+ n c)
          when (>= (+ (aref dp1 (max 0 end1))
                      (aref dp2 (min n end2)))
                   k)
          do (incf (aref counts (max 0 end1)))
             (decf (aref counts (min n end2))))
    (dotimes (i n)
      (incf (aref counts (+ i 1)) (aref counts i)))
    (with-buffered-stdout
      (loop for i from 0 below n
            when (and (char= #\o (aref s i))
                      (zerop (aref counts i)))
            do (println (+ i 1))))))

#-swank (main)

;;;
;;; Test and benchmark
;;;

#+swank
(defun io-equal (in-string out-string &key (function #'main) (test #'equal))
  "Passes IN-STRING to *STANDARD-INPUT*, executes FUNCTION, and returns true if
the string output to *STANDARD-OUTPUT* is equal to OUT-STRING."
  (labels ((ensure-last-lf (s)
             (if (eql (uiop:last-char s) #\Linefeed)
                 s
                 (uiop:strcat s uiop:+lf+))))
    (funcall test
             (ensure-last-lf out-string)
             (with-output-to-string (out)
               (let ((*standard-output* out))
                 (with-input-from-string (*standard-input* (ensure-last-lf in-string))
                   (funcall function)))))))

#+swank
(defun get-clipbrd ()
  (with-output-to-string (out)
    (run-program "powershell.exe" '("-Command" "Get-Clipboard") :output out :search t)))

#+swank (defparameter *this-pathname* (uiop:current-lisp-file-pathname))
#+swank (defparameter *dat-pathname* (uiop:merge-pathnames* "test.dat" *this-pathname*))

#+swank
(defun run (&optional thing (out *standard-output*))
  "THING := null | string | symbol | pathname

null: run #'MAIN using the text on clipboard as input.
string: run #'MAIN using the string as input.
symbol: alias of FIVEAM:RUN!.
pathname: run #'MAIN using the text file as input."
  (let ((*standard-output* out))
    (etypecase thing
      (null
       (with-input-from-string (*standard-input* (delete #\Return (get-clipbrd)))
         (main)))
      (string
       (with-input-from-string (*standard-input* (delete #\Return thing))
         (main)))
      (symbol (5am:run! thing))
      (pathname
       (with-open-file (*standard-input* thing)
         (main))))))

#+swank
(defun gen-dat ()
  (uiop:with-output-file (out *dat-pathname* :if-exists :supersede)
    (format out "")))

#+swank
(defun bench (&optional (out (make-broadcast-stream)))
  (time (run *dat-pathname* out)))

;; To run: (5am:run! :sample)
#+swank
(it.bese.fiveam:test :sample
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "11 3 2
ooxxxoxxxoo
"
    "6
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "5 2 3
ooxoo
"
    "1
5
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "5 1 0
ooooo
"
    "
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "16 4 3
ooxxoxoxxxoxoxxo
"
    "11
16
")))


Measured memory_bytes: 36584